


迷路を作るには



1

迷路を作る方法

よく使われる方法は2しゅるい

(1) 棒たおし法

まず迷路の外がわのかべを作る。中に1つおきに「棒」を立てて、ランダムに上下左右にたおして、かべにする。

- 《利点》プログラムが単純。迷路を作る時間もかからない。
- 《欠点》正解の道が直線になりやすい&片方のかべに寄りやすい。解く人にはわかりやすい迷路になってしまう。

(2) 穴ほり法

迷路を全部かべで埋めた後、穴をほって道を作る。ほり進められなくなったら、道をもどって、違う方向へ分かれ道をほる。


- 《利点》迷路を作る方法としてわかりやすい。正解の道がかなりバラバラになり、解く人には難しい迷路になる。
- 《欠点》プログラムが複雑。迷路を作るのに時間がかかる。

2


●棒たおし法の様子





(1) 外側のかべを作る



(2) 1つおきに棒を立てる



(3) 棒を上下左右へランダムにたおす



(4) 棒を全部たおすと完成

3




●穴ほり法の様子



(1) 全部をかべにする



(2) スタートから道をほる



(3) 行き止まったらもどって分かれ道 (4) 分かれ道を作れなくなったら完成

4


棒たおし法

5

まわりのかべを作る

```
10 CLV:CLS      変数クリア:画面クリア
20 FOR X=0 TO 30
30 ?CHR$(1);    上のかべ
40 NEXT
50 FOR Y=1 TO 21
60 LC 0,Y: ?CHR$(1)
70 LC 30,Y: ?CHR$(1)  左右のかべ
80 NEXT
90 FOR X=0 TO 30
100 ?CHR$(1);      下のかべ
110 NEXT
300 CONT        そのまま止まる
```

4方向のかべを表示
ESCキーを押すと終了



6

1つおきに棒を立てる

```


120 FOR Y=2 TO 20 STEP 2
130 FOR X=2 TO 28 STEP 2
140 LC X,Y: ?CHR$(1) 棒を立てる
200 NEXT
210 NEXT

```

Y方向に1つおきにくりかえし

X方向に1つおきにくりかえし

1つおきに棒を表示



7

棒を4方向にたおす

```

120 FOR Y=2 TO 20 STEP 2
130 FOR X=2 TO 28 STEP 2
140 LC X,Y: ?CHR$(1)
145 IF !INKEY() CONT 棒を立てた後にキー入力待
150 R=RND(4) 乱数で上下左右のどちらか
160 A=0: B=0
170 IF R=0 A=-1
171 IF R=1 A=1
172 IF R=2 B=-1
173 IF R=3 B=1
175 IF SCR(X+A,Y+B) GOTO 150
180 LC X+A,Y+B: ?CHR$(1) 棒を
190 IF !INKEY() CONT たおす
200 NEXT
210 NEXT

```

棒を立てた後にキー入力待

乱数で上下左右のどちらか


4方向それぞれのたおす位置を設定

その方向にすでに棒があったら、乱数出し直し

棒を
たおす

キー入力待

キーを押していくと
迷路ができる



8


棒をたおす方向をちょうせい

棒の1行目は上下左右、
2行目からは左右+下のみ

```
170 IF R=0 A=-1  
171 IF R=1 A=1  
172 IF R=2 IF Y=2 B=-1 ELSE GOTO 150  
173 IF R=3 B=1
```

棒の1行目は上にたおすが、2行目からはたおさない

正解が一通りの
迷路ができる




9

キー入力待ちをやめる

```
145 'IF !INKEY() CONT  
190 'IF !INKEY() CONT
```

先頭に「'」を付けて
コメントにする

キーを押さなくても
迷路ができる



10

ゲームにする

```

300 X=1:Y=1:LC 29,21:?"G":CLT
310 U=X-BTN(28)+BTN(29)
320 V=Y-BTN(30)+BTN(31)
330 LC 0,23:?"TICK()/60;
340 C=SCR(U,V)
350 IF C=1 GOTO 310
360 LC X,Y:?"CHR$(0)
370 X=U:Y=V
380 LC X,Y:?"CHR$(249)
390 WAIT 6
400 IF C!=71 GOTO 310
410 LC 0,0:CLK
420 END

```

キャラをスタート位置に:ゴール表示:タイムリセット

キャラを左右移動

キャラを上下移動

画面左下にタイムを表示

行き先がかべだったら移動しない

キャラを消す

キャラの位置を更新

キャラを表示

時間待ち


ゴールに着いていなかったら続ける

カーソルを左上へ:キーバッファをクリア

ゲーム終了

11

ゲームにする



人間のキャラを上下左右の矢印キーで操作
右下のゴールへ着いたら終了
画面左下にタイム(秒数)が出る

12

棒たおし法 完成(1)

```
10 CLV:CLS
20 FOR X=0 TO 30
30 ?CHR$(1);
40 NEXT
50 FOR Y=1 TO 21
60 LC 0,Y: ?CHR$(1)
70 LC 30,Y: ?CHR$(1)
80 NEXT
90 FOR X=0 TO 30
100 ?CHR$(1);
110 NEXT
120 FOR Y=2 TO 20 STEP 2
130 FOR X=2 TO 28 STEP 2
140 LC X,Y: ?CHR$(1)
145 /IF !INKEY() CONT
150 R=RND(4)
160 A=0:B=0
170 IF R=0 A=-1
171 IF R=1 A=1
```


13

棒たおし法 完成(2)

```
172 IF R=2 IF Y=2 B=-1 ELSE GOTO 150
173 IF R=3 B=1
175 IF SCR(X+A,Y+B) GOTO 150
180 LC X+A,Y+B: ?CHR$(1)
190 /IF !INKEY() CONT
200 NEXT
210 NEXT300 X=1:Y=1:LC 29,21: ?"G":CLT
310 U=X-BTN(28)+BTN(29)
320 V=Y-BTN(30)+BTN(31)
330 LC 0,23: ?TICK()/60;
340 C=SCR(U,V)
350 IF C=1 GOTO 310
360 LC X,Y: ?CHR$(0)
370 X=U:Y=V
380 LC X,Y: ?CHR$(249)
390 WAIT 6
400 IF C!=71 GOTO 310
410 LC 0,0:CLK
420 END
```

14

棒たおし法・グラフィック版



IchigoJam1.4のグラフィック機能で、もっと細かい迷路を作る。
ゲームのキャラは点滅するドットで表示。

15

棒たおし法・グラフィック版(1)

```

10 CLV:CLS
20 DRAW 0,0,62,0
30 DRAW 0,0,0,44
40 DRAW 62,0,62,44
50 DRAW 0,44,62,44
120 FOR Y=2 TO 20 STEP 2
130 FOR X=2 TO 28 STEP 2
140 DRAW X,Y
145 /IF !INKEY() CONT
150 R=RND(4)
160 A=0:B=0
170 IF R=0 A=-1
171 IF R=1 A=1
172 IF R=2 IF Y=2 B=-1 ELSE GOTO 150
173 IF R=3 B=1
175 IF POINT(X+A,Y+B) GOTO 150
180 DRAW X+A,Y+B
190 /IF !INKEY() CONT
200 NEXT
210 NEXT

```

上下左右のかべを表示

棒を表示

すでにかべだったら、そちらにはたおさない

棒をたおしてかべにする

16

棒たおし法・グラフィック版(2)

```

300 X=1:Y=1:CLT   キャラをスタート位置に:タイムリセット
305 DRAW X,Y,0    キャラのドットを消す
310 U=X-BTN(28)+BTN(29)
320 V=Y-BTN(30)+BTN(31)
325 DRAW X,Y,1    キャラのドットを表示
330 LC 0,23:7TICK()/60;
340 C=POINT(U,V)  行き先にある物をチェック
350 IF C GOTO 310  行き先がかべだったら移動しない
360 DRAW X,Y,0    キャラのドットを消す
370 X=U:Y=V
380 DRAW X,Y,1    キャラのドットを表示
390 WAIT 6        ゴールに着いていなかったらゲーム続行
400 IF X!=61 || Y!=43 GOTO 305
410 LC 0,0:CLK:OK 2
420 END           OK表示をOFF(終了時のスクロール防止)

```

17


穴ほり法

18

かべを作る

```
10 CLV:CLS      変数クリア:画面クリア
20 FOR Y=0 TO 22 Y方向くりかえし
30 FOR X=0 TO 30 X方向くりかえし
40 ?CHR$(1);    かべを表示
50 NEXT:        行の右はじで改行
60 NEXT
300 CONT        そのまま止まる
```

全部をかべでうめる
ESCキーを押すと終了




19

左上から道をほりはじめる

```
70 X=1:Y=1      スタート位置を左上に
80 LC X,Y:CHR$(0) 道をほる
```

左上に道がほられる



20

道をほり進める


```

100 D=RND(4):R=0 乱数で4方向のどれか:方向カウンターをリセット
110 GSB 500       チェックサブルーチンを呼ぶ
120 IF R=0 GOTO 100 その方向へ掘れなかったら別の方向へ
130 LC X+A,Y+B:?CHR$(0) 1歩目を掘る
140 LC X+A*2,Y+B*2:?CHR$(0) 2歩目を掘る
150 X=U:Y=V       掘り位置を更新
160 IF !INKEY() CONT キーが押されるまで待つ
170 GOTO 100      もどってくりかえし
500 A=0:B=0       ↓以下、チェックサブルーチン↓
510 IF D=0 A=-1   4方向の変化量を設定
520 IF D=1 A=1
530 IF D=2 B=-1
540 IF D=3 B=1
550 U=X+A*2:V=Y+B*2 次の位置を計算
560 IF U<0 OR U>30 RTN 横方向にはみ出していたらもどる
570 IF V<0 OR V>22 RTN 縦方向にはみ出していたらもどる
580 IF SCR(U,V)=0 RTN 次の位置が道だったらもどる
590 R=R+1         方向カウンターを+1
600 RTN          もどる

```

21

道をほる



キーを押すと、行き止まりになるところまで道をほる

22

掘る前に分かれ道ができる場所を記憶／取り出し

```

90 R=0:FOR D=0 TO 3
91 GSB 500
92 NEXT
95 IF R=0 X=[P]%100:Y=[P]/100:P=P-1:
GOTO 90
97 IF R>1 P=P+1:[P]=X+Y*100
170 GOTO 90

```

道を掘れる方向をカウント

行き止まりなら、直前の分かれ道ポイントへもどって掘りなおし


2方向以上に掘れるなら、今の場所を配列変数に記憶

もどる行を変更


掘りながら、分かれ道が作れる場所を配列変数に記憶
行き止まりになったら、配列変数から直前の場所を読み出してもどる

23

ほりながら分かれ道ができる場所を記憶



行き止まりになったら
分かれ道を作る



最後まで掘る

24

分かれ道を作れなくなったら終了

行き止まりで、分かれ道ポイントがもう無いなら300行へ

```
95 IF R=0 IF P=1 GOTO 300 ELSE X=[
P]%100:Y=[P]/100:P=P-1:GOTO 90
```


```
160 / IF !INKEY() CONT
```

キー入力待ちをコメントにする

```
300 CONT
```

そのまま待つ

分かれ道が作れなくなったら
完成。
300行へジャンプして止まる。



25

ゲームにする


(棒たおし法のゲームと同じ)

キャラをスタート位置に:ゴール表示:タイムリセット

```
300 X=1:Y=1:LC 29,21:"G":CLT
310 U=X-BTN(28)+BTN(29) キャラを左右移動
320 V=Y-BTN(30)+BTN(31) キャラを上下移動
330 LC 0,23:"TICK()/60" 画面左下にタイムを表示
340 C=SCR(U,V)
350 IF C=1 GOTO 310 } 行き先がかべだったら
                      } 移動しない
360 LC X,Y:"CHR$(0) キャラを消す
370 X=U:Y=V キャラの位置を更新
380 LC X,Y:"CHR$(249) キャラを表示
390 WAIT 6 時間待ち
400 IF C!=71 GOTO 310 ゴールに着いていなかったら続ける
410 LC 0,0:CLK カーソルを左上へ:キーバッファをクリア
420 END ゲーム終了
```

26

ゲームにする




人間のキャラを上下左右の矢印キーで操作
右下のゴールへ着いたら終了
画面左下にタイム(秒数)が出る

27

もっと難しい迷路にする

70 X=29:Y=1

掘り始める位置を右上にする



ここから
掘りはじめる

28

穴ほり法 完成(1)

```

10 CLV:CLS
20 FOR Y=0 TO 22
30 FOR X=0 TO 30
40 ?CHR$(1);
50 NEXT:
60 NEXT
70 X=29:Y=1
80 LC X,Y: ?CHR$(0)
90 R=0:FOR D=0 TO 3
91 GSB 500
92 NEXT
95 IF R=0 IF P=1 GOTO 300 ELSE X=[P]
%100:Y=[P]/100:P=P-1:GOTO 90
97 IF R>1 P=P+1:[P]=X+Y*100

```

29

穴ほり法 完成(2)

```

100 D=RND(4):R=0
110 GSB 500
120 IF R=0 GOTO 100
130 LC X+A,Y+B: ?CHR$(0)
140 LC X+A*2,Y+B*2: ?CHR$(0)
150 X=U:Y=V
160 /IF !INKEY() CONT
170 GOTO 90
300 X=1:Y=1:LC 29,21: ?"G":CLT
310 U=X-BTN(28)+BTN(29)
320 V=Y-BTN(30)+BTN(31)
330 LC 0,23: ?TICK()/60;
340 C=SCR(U,V)
350 IF C=1 GOTO 310
360 LC X,Y: ?CHR$(0)
370 X=U:Y=V
380 LC X,Y: ?CHR$(249)
390 WAIT 6
400 IF C!=71 GOTO 310
410 LC 0,0:CLK
420 END

```


30

穴ほり法 完成(3)

```
500 A=0:B=0
510 IF D=0 A=-1
520 IF D=1 A=1
530 IF D=2 B=-1
540 IF D=3 B=1
550 U=X+A*2:V=Y+B*2
560 IF U<0 OR U>30 RTN
570 IF V<0 OR V>22 RTN
580 IF SCR(U,V)=0 RTN
590 R=R+1
600 RTN
```

31

穴ほり法・グラフィック版



IchigoJam1.4のグラフィック機能で、もっと細かい迷路を作る。
ゲームのキャラは点滅するドットで表示。

32

穴ほり法・グラフィック版(1)

```

10 CLV:CLS
20 FOR Y=0 TO 44 }
30 DRAW 0,Y,62,Y } 全部をかべでうめる
40 NEXT
60 P=#6FE        分かれ道メモリを初期設定
70 X=61:Y=1      ほり始める位置を右上に
80 DRAW X,Y,0     スタート位置を道にする
90 GSB 200        カウントサブルーチンを呼ぶ
95 IF R=0 IF P<=#700 GOTO 300 ELSE G
96 700:GOTO 90
97 IF R>1 P=P+2:POKE P,X,Y 2方向以上にほれるなら
                              今の場所をメモリに記憶
100 D=RND(4):R=0
110 GSB 500
120 IF R=0 GOTO 100
130 DRAW X+A,Y+B,X+A*2,Y+B*2,0 道をほる
150 X=U:Y=V
160 /IF !INKEY() CONT
170 GOTO 90

```

33

穴ほり法・グラフィック版(2)

```

200 R=0:FOR D=0 TO 3 } 分かれ道カウントサブルーチン
210 GSB 500 } 全4方向で分かれ道が何本作れるか
220 NEXT } カウント
230 RTN
300 X=1:Y=1:CLT      キャラをスタート位置に:タイムリセット
305 DRAW X,Y,0       キャラのドットを消す
310 U=X-BTN(28)+BTN(29)
320 V=Y-BTN(30)+BTN(31)
325 DRAW X,Y,1       キャラのドットを表示
330 LC 0,23:7TICK()/60;
340 C=POINT(U,V)     行き先にある物をチェック
350 IF C GOTO 310     行き先がかべだったら移動しない
360 DRAW X,Y,0       キャラのドットを消す
370 X=U:Y=V
380 DRAW X,Y,1       キャラのドットを表示
390 WAIT 6           ゴールに着いていなかったらゲーム続行
400 IF X!=61 OR Y!=43 GOTO 305
410 LC 0,0:CLK:OK 2
420 END              OK表示をOFF(終了時のスクロール防止)

```

34

穴ほり法・グラフィック版(3)

```

500 A=0:B=0
510 IF D=0 A=-1
520 IF D=1 A=1
530 IF D=2 B=-1
540 IF D=3 B=1
550 U=X+A*2:V=Y+B*2
560 IF U<0 OR U>62 RTN 横方向にはみ出していたらもどる
570 IF V<0 OR V>44 RTN 縦方向にはみ出していたらもどる
580 IF POINT(U,V)=0 RTN 次の位置が道だったらもどる
590 R=R+1
600 RTN ↓分かれ道メモリ処理サブルーチン↓ 分かれ道がほれないメモリをチェック
                                                    そこをつめてメモリを節約
700 Q=#700
710 X=PEEK(Q):Y=PEEK(Q+1):GSB 200
720 IF !R COPY Q,Q+2,P-Q:P=P-2:IF
Q<P GOTO 710
730 IF Q<P Q=Q+2:GOTO 710
740 Q=#700:B=#FFE&RND(P-Q+2):A=Q+B
750 X=PEEK(A):Y=PEEK(A+1)
760 RTN 分かれ道がほれるメモリからランダムに選んで、ほる位置を移動

```

35